Fermat: Ein absolut namenloses Beispiel

Zunächst müssen wir mal den Speicher entleeren und nötige Pakete laden. Das braucht Dich nicht zu interessieren...

```
> restart; with
> (plots):
```

```
Warning, the name changecoords has been redefined
```

Und los! Wir wissen daß die Durchlaufzeit einer Kurve $\mathrm{y}(x)$ in einem Lichtmedium, in dem die Lichtgeschwindigkeit c(x,y) vom Ort abhängt, gegeben ist durch:

$$\mathrm{T}(y) = \int_{x_1}^{x_2} \frac{\sqrt{1+(\frac{d}{dx}\,\mathrm{y}(x))^2}}{\mathrm{c}(x,\,y)}\,dx$$

Nehmen wir zum Beispiel: (Probiere doch auch mal was Anderes aus!)

```
> c:= (x,y)-> exp(y);
```

$$c := (x,\,y) \to e^y$$

Wir haben also als Funktion F:

```
> F:= (u,v,w) -> sqrt(1+v^2)/c(w,u);
```

$$F := (u,\,v,\,w) \to \frac{\sqrt{1+v^2}}{\mathrm{c}(w,\,u)}$$

Jetzt lassen wir Maple die partielle Ableitung nach dem ersten Argument von F berechnen. Müssen wir Maple etwas überreden, um uns das Ergebnis auch zuzumuten.

```
> D[1](F)(u,v,w);
> F1:= unapply( diff(F(u,v,w),u), u,v,w);
```

$$\mathrm{D}_1(F)(u,\,v,\,w)$$

$$F1 := (u,\,v,\,w) \to -\frac{\sqrt{1+v^2}}{e^u}$$

Jetzt brauchen wir das für die Euler-Langrange-Gleichung an der Stelle ($\mathrm{y}(x)$, $\frac{d}{dx}\,\mathrm{y}(x)$, x).

```
> term1:=F1( y(x), D(y)(x), x);
```

$$term1 := -\frac{\sqrt{1+\mathrm{D}(y)(x)^2}}{e^{\mathrm{y}(x)}}$$

Jetzt zum zweiten Teil der Euler-Lagrange-Gleichung. Wieder haben wir ein wenig mit Maple zu kämpfen.

```
> F2:= unapply( diff(F(u,v,w),v), u,v,w);
> term2:=diff ( F2( y(x), D(y)(x), x) , x);
```

$$F2 := (u,\,v,\,w) \to \frac{v}{\sqrt{1+v^2}\,e^u}$$

$$term2 := -\frac{\mathrm{D}(y)(x)^2\,(\mathrm{D}^{(2)})(y)(x)}{(1+\mathrm{D}(y)(x)^2)^{(3/2)}\,e^{\mathrm{y}(x)}} - \frac{\mathrm{D}(y)(x)\,(\frac{d}{dx}\,\mathrm{y}(x))}{\sqrt{1+\mathrm{D}(y)(x)^2}\,e^{\mathrm{y}(x)}} + \frac{(\mathrm{D}^{(2)})(y)(x)}{\sqrt{1+\mathrm{D}(y)(x)^2}\,e^{\mathrm{y}(x)}}$$

Puh, das sieht ganz schön kompliziert aus. Aber was haben wir denn auch anderes erwartet! Ein Glück, daß wir Maple haben! Dann stellen wir mal die Euler-Lagrange-Gleichung auf! Der

```
> simplify
```

-Befehl hilft uns, das Ganze etwas einfacher darstellen zu lassen!

```
> simplify(term1-term2=0);
```

$$-\frac{(1+\mathrm{D}(y)(x)^2+(\mathrm{D}^{(2)})(y)(x))\,e^{(-\mathrm{y}(x))}}{(1+\mathrm{D}(y)(x)^2)^{(3/2)}}=0$$

Oha, das ist wirklich eine Verbesserung! Jetzt wird es spannend! Vielleicht haben wir Glück, und Maple findet eine Lösung. Wer verwenden den

```
> dsolve
```

-Befehl zum Auflösen der Differentialgleichung nach y(x):

```
> dsolve(term1-term2=0, y(x));
```

$$\mathrm{y}(x)=\ln(-\_C1\,\sin(x)+\_C2\,\cos(x))$$

Jetzt zeigt Maple veschiedene Möglichkeiten für die Lösungen an. Welches nun wirklich die richtige Lösung ist, muß man sich anderweitig überlegen. _C1 und _C2 sind nur Integrationskonstanten. Man muß sie aus den übrigen Bedingungen bestimmen.

Dann wollen wir mal versuchen, die Integrationskonstanten zu bestimmen. Zunächst geben wir der allgemeinen Lösung einen Namen:

```
> y1:= x->ln(-C1*sin(x)+C2*cos(x));
```

Nun kann man irgendwelche Bedingungen einsetzen, z.B. soll $\mathrm{y1}(3)=4$ und $\mathrm{y1}(2)=1$ gelten. Diese Gleichungen kann man so schreiben...

```
> glg1:= y1(0)=4;
> glg2:= y1(2)=5;
```

$$y1 := x \to \ln(-C1\,\sin(x)+C2\,\cos(x))$$

$$glg1 := \ln(C2)=4$$

$$glg2 := \ln(-C1\,\sin(2)+C2\,\cos(2))=5$$

... und Maple löst sie sogar für uns!

```
> C2:=solve(glg1, C2);
> C1:=solve(glg2, C1);
```

$$C2 := e^4$$

$$C1 := \frac{-e^5+e^4\cos(2)}{\sin(2)}$$

Und schließlich plotten wir das mal!

```
> plot( y1(x), x=0..4, scaling=constrained);
```


5
4
3
2
1
0
–1
–2
1
2
3
4
x